/*
   Copyright 2008, Andres Postiglioni

   Licensed under the Apache License, Version 2.0 (the "License");
   you may not use this file except in compliance with the License.
   You may obtain a copy of the License at

       http://www.apache.org/licenses/LICENSE-2.0

   Unless required by applicable law or agreed to in writing, software
   distributed under the License is distributed on an "AS IS" BASIS,
   WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
   See the License for the specific language governing permissions and
   limitations under the License.
*/
package com.aconcagua.beans.support {
  import com.aconcagua.beans.BeanDefinition;
  import com.aconcagua.beans.BeanDefinitionRegistrar;
  import com.aconcagua.factory.BeanFactory;

  import mx.core.IMXMLObject;

  /**
   * @author Andres Postiglioni
   */
  [DefaultProperty("metadata")]
  public class BeanDefinitionRegistrarSupport implements BeanDefinitionRegistrar, IMXMLObject {
    private var _id: String;
    private var _metadata: Array;

    private var _bean: BeanDefinition;
    private var _factoryMethod: Function;

    public function BeanDefinitionRegistrarSupport(factoryMethod: Function) {
      _factoryMethod = factoryMethod;
    }

    public function initialized(document:Object, id:String): void {
      _id = id;
    }

    public function get id(): String {
      return _id;
    }

    public function set metadata(meta: Array): void {
      _metadata = meta;
    }

    public function registerAlias(name: String, beanFactory: BeanFactory): void {
      if (!isRegisteredIn(beanFactory)) {
        this.registerIn(beanFactory);
      }

      beanFactory.registerAlias(name, _bean);
    }

    public function isRegisteredIn(beanFactory: BeanFactory): Boolean {
      return _bean != null && beanFactory.containsBean(_bean.name);
    }

    public function registerIn(beanFactory: BeanFactory): void {
      _bean = this.createBeanDefinition(_id, _metadata, beanFactory);
      beanFactory.registerBeanDefinition(_bean);
    }

    public function retrieveFrom(beanFactory: BeanFactory): Object {
      return beanFactory.getBean(_bean.name);
    }

    protected function createBeanDefinition(id: String, metadata: Array, beanFactory: BeanFactory): BeanDefinition {
      return _factoryMethod.apply(this, [id, metadata]);
    }
  }
}